/****************************************************************************

%VARIANCE macro: Calculating intra-class correlation (ICC) and reliability

DISCLAIMER:
	This information is provided by Center for Healthy Aging, Pennsylvania
State University as a service to researchers/SAS users who address repeated assessment data. It is provided "AS IS". There are no warranties, expressed or implied, as to merchantability or fitness for a particular purpose regarding the accuracy of the materials or code contained herein.

PURPOSE:
a. The %VARIANCE macro generates within- and between-person variances in multilevel model and calculates ICC with the variances.

b. The macro calculates within- and between-person reliabilities of scale (e.g., assessing negative/positive affect [NA/PA] in this example code) in multilevel model. The within- and between-person reliabilities were calculated following Cranford et al. (2006).

c. Reliability calculations were also simulated with various number of items in order to answer questions regarding how many items are needed to construct a reliable measure of differences between occasions within the same person and stable differences between individuals to achieve a particular level of reliability. We followed the examples provided by Segerstrom and colleagues (2014).

	Reference: 
1. Cranford, J. A., Shrout, P. E., Iida, M., Rafaeli, E., Yip, T., & Bolger, N. (2006). A Procedure for Evaluating Sensitivity to Within person Change: Can Mood Measures in Diary Studies Detect Change Reliably? Personality and Social Psychology Bulletin, 32(7), 917–929. https://doi.org/10.1177/0146167206287721
2. Segerstrom, S. C., Boggero, I. A., Smith, G. T., & Sephton, S. E. (2014). Variability and Reliability of Diurnal Cortisol in Younger and Older Adults: Implications for Design Decisions. Psychoneuroendocrinology, 49(Supplement C), 299–309. https://doi.org/10.1016/j.psyneuen.2014.07.022

REQUIRES:
The %VARIANCE macro was written using Version 9.4. It may require Version 9.4 or later of SAS software to run this macro.

USAGE:
	The macro input parameters are:
DATASET = Name of "dataset" contains information participant ID, 
day in study, and sub-items of scale. Macro works with more than a single dataset (typically expects multiple datasets).
 
TYPE = Study type of each DATASET. Macro expects to identify 
"EMA" or "EOD" for each DATASET. EMA stands for Ecological Momentary Assessment which has multiple observations within a day. EOD (e.g., similar study type to daily dairy method) stands for End of Day assessment which has a single observation within a day but multiple observations across the study period.

        	ID = Variable name of "participants ID" in each DATASET.

        	STUDYDAY = Variable name of "day in study" in each DATASET.

SCALE = Scale name (e.g., NA [negative affect] or PA [positive 
affect]).

		DATASET_TYPE_SCALE = Sub-items for SCALE in each dataset.

END = The maximum number of items you want to consider in order 
to answer the questions regarding how many items are needed to construct a reliable measure of differences between occasions within the same person and stable differences between individuals to achieve a particular level of reliability. The number of items ranging from two to "END" is tested.

LIMITATIONS:
No error checking is fully done. The macro assumes that all parameters above entered are valid. The calculation of reliability can generate a memory error with a large sample (e.g., n>800). Simple random sample (SRS) or MEMSIZE option can be used to solve the issue by decreasing the sample size or increasing memory usage for SAS, respectively. However, the solutions are not guaranteed.

EXAMPLE:
This example code imports 4 EOD and 5 EMA datasets and calculates ICC and within-/between-person reliabilities of scale assessing NA or PA in multilevel model.

*Specifying all datasets for macro function (Datasets should be listed with one space between words);

		%let DATASET = NSDE WFHS ESCAPE SAWM SHADE NTH WDL ESCAPE SAWM;

*Specifying the study type of each dataset (e.g., EOD or EMA). The order of the type should follow the order of DATASET (Types should be listed with one space between words);

		%let TYPE = EOD EOD EOD EOD EMA EMA EMA EMA EMA;

*IMPORTANT: The filename of datasets in the folder you will specify should be given by "DATASET_TYPE.sas7bdat". For example, "NSDE_EOD.sas7bdat";

*Setting key variables (i.e., participant ID and day in study) in each dataset for macro function. The order of each variable should follow the order of DATASET (variables should be listed with one space between words);

		%let ID = M2ID EID ID ID ID ID ID ID ID;

%let STUDYDAY = b2dday eday studyday studyday EMA_day EMA_day 
studyday studyday studyday;

*Specifying scale names (e.g., NA or PA) and their items you want to test in each dataset (the scale names and their items should be listed with one space between words). Scale names can be arbitrarily chosen, but the items should be included in each dataset with the same variable name. Simplified variable names are provided in this example code. Also, the name of the parameter "DATASET_TYPE_SCALE" should be listed with the actual name of DATASET, TYPE, and SCALE, e.g., "NSDE_EOD_NA";

		%let SCALE = NA PA;

	*End of Day (EOD);

%let NSDE_EOD_NA = nervous jittery sad_nothing_could_cheer_you_up 
lonely hopeless worthless frustrated angry irritable afraid ashamed restless_or_fidgety everything_was_an_effort upset;
%let NSDE_EOD_PA = extemely_happy enthusiastic in_good_spirits 
calm_and_peaceful satisfied cheerful full_of_life close_to_others like_you_belong attentive proud active confident;
		%let WFHS_EOD_NA = nervous jittery distressed hostile irritable 
afraid ashamed scared guilty upset;
		%let WFHS_EOD_PA = enthusiastic excited attentive proud active 
interested strong alert inspired determined;
		%let ESCAPE_EOD_NA = tense depressed unhappy frustrated angry;
		%let ESCAPE_EOD_PA = happy pleased enjoyment joyful;
		%let SAWM_EOD_NA = tense sad disappointed upset;
		%let SAWM_EOD_PA = happy enthusiastic excited content;

	*Ecological Momentary Assessement (EMA);

		%let SHADE_EMA_NA = worried depressed unhappy frustrated angry;
		%let SHADE_EMA_PA = happy pleased enjoyment joyful;
		%let NTH_EMA_NA = tense nervous depressed sad angry hostile;
		%let NTH_EMA_PA = happy calm relaxed cheerful lively energetic;
		%let WDL_EMA_NA = sad;
		%let WDL_EMA_PA = happy;
		%let ESCAPE_EMA_NA = tense depressed unhappy frustrated angry;
		%let ESCAPE_EMA_PA = happy pleased enjoyment joyful;
		%let SAWM_EMA_NA = tense sad disappointed upset;
		%let SAWM_EMA_PA = happy enthusiastic excited content;

*Specifying the maximum number of items you want to test for the PURPOSE c. The number of items ranging from two to nine (e.g., if END=9) will be considered in the following procedure.

		%let END = 9;

****************************************************************************/




[bookmark: _GoBack]*This example code imports 4 EOD and 5 EMA datasets and calculates ICC and within-/between-person reliabilities of scale assessing NA or PA in multilevel model;

/*****************************************************************************
* 	<Importing 9 datasets>                                                 *
*   	4 EOD – NSDE, WFHS, ESCAPE, SAWM                                       *
*   	5 EMA – SHADE, NTH, WDL, ESCAPE, SAWM                                  *
*****************************************************************************/

*Specifying path to the datasets;
libname data "C:\Users\datasets";

*Specifying all datasets for macro function (Datasets should be listed with one space between words);
%let Dataset = NSDE WFHS ESCAPE SAWM SHADE NTH WDL ESCAPE SAWM;
*Specifying the study type of each dataset (e.g., EOD or EMA). The order of the type should follow the order of Dataset you set above (Types should be listed with one space between words);
%let Type = EOD EOD EOD EOD EMA EMA EMA EMA EMA; 

*IMPORTANT: The filename of datasets in the folder you specified above should be given by "Dataset_Type.sas7bdat". For example, "NSDE_EOD.sas7bdat".

/*****************************************************************************
*	<Specifying key variables in each dataset Importing 9 datasets>        *
*****************************************************************************/

*Setting key variables in each dataset for macro function. The order of each variable should follow the order of Dataset you set above (variables should be listed with one space between words);
%let ID = M2ID EID ID ID ID ID ID ID ID;
%let Studyday = b2dday eday studyday studyday EMA_day EMA_day studyday studyday studyday;

*Specifying scale names (e.g., NA or PA) and their items you want to test in each dataset (the scale names and their items should be listed with one space between words). Scale names can be arbitrarily chosen, but the items should be included in each dataset with the same variable name. Simplified variable names are provided in this example code. Also, the name of the parameter "DATASET_TYPE_SCALE" should be listed with the actual name of DATASET, TYPE, and SCALE, e.g., "NSDE_EOD_NA";

%let Scale = NA PA;

*End of Day (EOD);
%let NSDE_EOD_NA = nervous jittery so_sad_nothing_could_cheer_you_up lonely hopeless worthless frustrated angry irritable afraid ashamed restless_or_fidgety everything_was_an_effort upset;
%let NSDE_EOD_PA = extemely_happy enthusiastic in_good_spirits calm_and_peaceful satisfied cheerful full_of_life close_to_others like_you_belong attentive proud active confident;
%let WFHS_EOD_NA = nervous jittery distressed hostile irritable afraid ashamed scared guilty upset;
%let WFHS_EOD_PA = enthusiastic excited attentive proud active interested strong alert inspired determined;
%let ESCAPE_EOD_NA = tense depressed unhappy frustrated angry;
%let ESCAPE_EOD_PA = happy pleased enjoyment joyful;
%let SAWM_EOD_NA = tense sad disappointed upset;
%let SAWM_EOD_PA = happy enthusiastic excited content;

*Momentary (EMA);
%let SHADE_EMA_NA = worried depressed unhappy frustrated angry;
%let SHADE_EMA_PA = happy pleased enjoyment joyful;
%let NTH_EMA_NA = tense nervous depressed sad angry hostile;
%let NTH_EMA_PA = happy calm relaxed cheerful lively energetic;
%let WDL_EMA_NA = sad;
%let WDL_EMA_PA = happy;
%let ESCAPE_EMA_NA = tense depressed unhappy frustrated angry;
%let ESCAPE_EMA_PA = happy pleased enjoyment joyful;
%let SAWM_EMA_NA = tense sad disappointed upset;
%let SAWM_EMA_PA = happy enthusiastic excited content;

*Specifying the maximum number of items you want to test for the PURPOSE c. The number of items ranging from two to nine (e.g., if END=9) will be considered in the following procedure;

%let END = 9;

/*****************************************************************************
*	<Code that produced ICC and reliabilities of scale assessing NA or     *
*	PA in multilevel model>                                                *
*****************************************************************************/

%macro Variance;

*Reading datasets and key variables one by one;
%do i=1 %to %eval(%sysfunc(countc(&Dataset.," "))+1);
   %let Dataset1 = %scan(&Dataset, &i);
   %put Dataset1;
   %let Type1 = %scan(&Type, &i);
   %put Type1;
   %let ID1 = %scan(&ID, &i);
   %put ID1;
   %let Studyday1 = %scan(&Studyday, &i);
   %put Studyday1;

%do j=1 %to %eval(%sysfunc(countc(&Scale.," "))+1);
   %let Scale1 = %scan(&Scale, &j);
   %put Scale1;

*Arranging data before the main analysis;
data tmp_scalemean; set data.&Dataset1._&Type1.;
*Calculating NA and PA from multiple items which used in each dataset;
&Scale1. = mean (of &&&dataset1._&Type1._&Scale1.);
*Specifying ID and Studyday in each dataset;
ID = &ID1.; Studyday = &Studyday1.;
run;
*Removing missing data;
data scalemean; set tmp_scalemean;
if &Scale1=. then delete;
run;

*Generating a PDF file for SAS results separately by dataset and scale;
ods pdf file= "C:\Users\Results\&Dataset1._&Type1._&Scale1..pdf";

*Calculating each person’s # of observation and person mean of NA/PA;
proc means data=scalemean nway noprint;
by ID;
var &Scale1.;
output out=&Dataset1.&Type1.&Scale1._pm
mean(&Scale1.)=&Scale1._pm;
run;
*Removing missing data;
data &Dataset1.&Type1.&Scale1._pm; set &Dataset1.&Type1.&Scale1._pm;
if &Scale1._pm=. then delete;
run;

*Calculating grand mean of # of observation and NA/PA;
proc means data=&Dataset1.&Type1.&Scale1._pm nway noprint;
var _FREQ_ &Scale1._pm;
output out=grand_mean
mean(_FREQ_)=n_obs
mean(&Scale1._pm)=&Scale1._gm;
run;

*Saving n_obs (i.e., Grand mean of # of observations from all participant) as macro variable for the next step;
proc sql noprint;
select n_obs
  into :n_obs
  from grand_mean;
quit;

*PURPOSE a. CALCULATING ICC: BETWEEN AND WITHIN PERSON VARIANCES IN NA AND PA;

title1 &Dataset1. &Type1.;
title2 &Scale1. Variance;
proc mixed data=scalemean noclprint noitprint covtest;
class id studyday;
model &Scale1.= /solution;
random intercept /subject=id type=un gcorr;
*Using 2-level or 3-level MLM according to study type (EMA vs. EOD); 
%if &type1.=EMA %then %do;
random intercept /subject=studyday(id); %end;
ods output CovParms=Variance_&Scale1.;
run;
*WP_day below is zero where study type = EOD;
proc sql;
create table ICC_&Scale1. as
select sum(estimate*(covparm="UN(1,1)"))   / sum(Estimate) as BP
      ,sum(estimate*(covparm="Intercept")) / sum(Estimate) as WP_day
      ,sum(estimate*(covparm="Residual"))  / sum(Estimate) as WP
from variance_&Scale1.;
quit;
proc print data=ICC_&Scale1.; run;

*PURPOSE b. CALCULATING RELIABILITY: WITHIN- AND BETWEEN-PERSON RELIABILITIES IN NA AND PA;

*Preparing dataset for reliability calculation;
*sorting data;
proc sort data=scalemean;
by ID Studyday;
run;
*making a variable "new_sess" that orders observations continuously within persons;
data obs_no; set scalemean;
by id;
if first.id then new_sess=0; new_sess+1;
run;
*creating a multivariate dataset with the items you want to correlate on distinct rows. This will allow you to calculate the extent to which these items covary across study;
%let item_no = %eval(%sysfunc(countc(&&&dataset1._&Type1._&Scale1., " "))+1);
*Conducting below only if NA or PA has more than 2 items (WDL was excluded from the analysis);
%if &item_no. ge 2 %then %do;

data gthry; set obs_no; 
by id studyday;
%do n=1 %to &item_no.;
   %let Scale_item = %scan(&&&dataset1._&Type1._&Scale1., &n);
   %put Scale_item;
y=&scale_item.; item=&n.; output;
%end;
run;
title1 &Dataset1. &Type1.; 
title2 Within and Between Person Reliabilities in &Scale1.;
proc varcomp data=gthry;
      class id new_sess item;
      model y= id new_sess item id*new_sess id*item new_sess*item;
	  ods output Estimates=reliability;
run;

/*****************************************************************************
* Within person reliability (within+across day calculation)                  * 
* How reliable from one obs to next across entire study; ignores day nesting *                                                     
* WPreliability = var(ID*New_sess)/[var(ID*New_sess)+(var(Error)/n_items)]   *
*****************************************************************************/

data wp_rel;
set reliability;
n_items=&Item_no.;
retain systematic;
if varcomp="Var(ID*new_sess)" | varcomp="Var(id*new_sess)" | varcomp="Var(Id*new_sess)" then systematic=y;
if varcomp="Var(Error)" then error=y/n_items;
wp_reliability=systematic/(systematic+error);
output;
run;
title WP reliability: &Dataset1. &Type1. &Scale1.;
proc print data=wp_rel;
where wp_reliability ne .;
var wp_reliability;
run;

/*****************************************************************************
* Between person reliability                                                 *
* Rkf: reliablity of a mood measure across all days                          *
* BPreliability = [var(ID)+(var(ID*Item)/n_items)]/                          *
*                [var(ID)+(var(ID*Item)/n_items)+var(Error)/(n_obs*n_items)] *
*****************************************************************************/

data bp_rel;
set reliability;
n_items=&Item_no.;
n_obs=&n_obs.;
retain per_var systematic error_kf;
if varcomp="Var(ID)" | varcomp="Var(id)" | varcomp="Var(Id)" then per_var=y;
if varcomp="Var(ID*new_sess)" | varcomp="Var(id*new_sess)" | varcomp="Var(Id*new_sess)" then systematic=y;
if varcomp="Var(Error)" then error_kf=y/(n_items*n_obs);
bp_reliability=(per_var+(systematic/n_items))/(per_var+(systematic/n_items)+error_kf);
output;
run;
title BP reliability: &Dataset1. &Type1. &Scale1.;
proc print data=bp_rel;
where bp_reliability ne .;
var bp_reliability;
run;

*PURPOSE c. SOLVING FOR # OF ITEMS: BETWEEN AND WITHIN PERSON RELIABILITES IN NA AND PA ACCORDING TO DIFFERENT # OF ITEMS;

*Simulating reliabilities with various # of items;
%do m=2 %to &End.;
*Within person reliability (within+across day calculation);
data wp_rel;
set reliability;
n_items=&m.;
retain systematic;
if varcomp="Var(ID*new_sess)" | varcomp="Var(id*new_sess)" | varcomp="Var(Id*new_sess)" then systematic=y;
if varcomp="Var(Error)" then error=y/n_items;
wp_reliability=systematic/(systematic+error);
output;
run;
data WP_&m.; set wp_rel(keep=wp_reliability);
if wp_reliability=. then delete;
n_items=&m.;
run;

*Between person reliability;
data bp_rel;
set reliability;
n_items=&m.;
n_obs=&n_obs.;
retain per_var systematic error_kf;
if varcomp="Var(ID)" | varcomp="Var(id)" | varcomp="Var(Id)" then per_var=y;
if varcomp="Var(ID*new_sess)" | varcomp="Var(id*new_sess)" | varcomp="Var(Id*new_sess)" then systematic=y;
if varcomp="Var(Error)" then error_kf=y/(n_items*n_obs);
bp_reliability=(per_var+(systematic/n_items))/(per_var+(systematic/n_items)+error_kf);
output;
run;
data BP_&m.; set bp_rel(keep=bp_reliability);
if bp_reliability=. then delete;
n_items=&m.;
run;

%end;

*Generating the table for reliabilities with various items;
data wp_item; set WP_2-WP_&End.; run;
title WP reliability: &Dataset1. &Type1. &Scale1. by # of item (2-&End.);
proc print data=wp_item;
run;
data bp_item; set BP_2-BP_&End.; run;
title BP reliability: &Dataset1. &Type1. &Scale1. by # of item (2-&End.);
proc print data=bp_item;
run;

%end;

ods pdf close;

%end;
%end;
%mend;

%Variance;


